// SPDX-License-Identifier: MIT
pragma solidity ^0.8.17;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
  /**
   * @dev Returns true if this contract implements the interface defined by
   * `interfaceId`. See the corresponding
   * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
   * to learn more about how these ids are created.
   *
   * This function call must use less than 30 000 gas.
   */
  function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
  /**
   * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
   */
  event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

  /**
   * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
   */
  event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

  /**
   * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
   */
  event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

  /**
   * @dev Returns the number of tokens in ``owner``'s account.
   */
  function balanceOf(address owner) external view returns (uint256 balance);

  /**
   * @dev Returns the owner of the `tokenId` token.
   *
   * Requirements:
   *
   * - `tokenId` must exist.
   */
  function ownerOf(uint256 tokenId) external view returns (address owner);

  /**
   * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
   * are aware of the ERC721 protocol to prevent tokens from being forever locked.
   *
   * Requirements:
   *
   * - `from` cannot be the zero address.
   * - `to` cannot be the zero address.
   * - `tokenId` token must exist and be owned by `from`.
   * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or
   * {setApprovalForAll}.
   * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a
   * safe transfer.
   *
   * Emits a {Transfer} event.
   */
  function safeTransferFrom(address from, address to, uint256 tokenId) external;

  /**
   * @dev Transfers `tokenId` token from `from` to `to`.
   *
   * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
   *
   * Requirements:
   *
   * - `from` cannot be the zero address.
   * - `to` cannot be the zero address.
   * - `tokenId` token must be owned by `from`.
   * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
   *
   * Emits a {Transfer} event.
   */
  function transferFrom(address from, address to, uint256 tokenId) external;

  /**
   * @dev Gives permission to `to` to transfer `tokenId` token to another account.
   * The approval is cleared when the token is transferred.
   *
   * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
   *
   * Requirements:
   *
   * - The caller must own the token or be an approved operator.
   * - `tokenId` must exist.
   *
   * Emits an {Approval} event.
   */
  function approve(address to, uint256 tokenId) external;

  /**
   * @dev Returns the account approved for `tokenId` token.
   *
   * Requirements:
   *
   * - `tokenId` must exist.
   */
  function getApproved(uint256 tokenId) external view returns (address operator);

  /**
   * @dev Approve or remove `operator` as an operator for the caller.
   * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
   *
   * Requirements:
   *
   * - The `operator` cannot be the caller.
   *
   * Emits an {ApprovalForAll} event.
   */
  function setApprovalForAll(address operator, bool _approved) external;

  /**
   * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
   *
   * See {setApprovalForAll}
   */
  function isApprovedForAll(address owner, address operator) external view returns (bool);

  /**
   * @dev Safely transfers `tokenId` token from `from` to `to`.
   *
   * Requirements:
   *
   * - `from` cannot be the zero address.
   * - `to` cannot be the zero address.
   * - `tokenId` token must exist and be owned by `from`.
   * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
   * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a
   * safe transfer.
   *
   * Emits a {Transfer} event.
   */
  function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;
}

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
  /**
   * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
   * by `operator` from `from`, this function is called.
   *
   * It must return its Solidity selector to confirm the token transfer.
   * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
   *
   * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
   */
  function onERC721Received(
    address operator,
    address from,
    uint256 tokenId,
    bytes calldata data
  )
  external
  returns (bytes4);
}

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
  /**
   * @dev Returns the token collection name.
   */
  function name() external view returns (string memory);

  /**
   * @dev Returns the token collection symbol.
   */
  function symbol() external view returns (string memory);

  /**
   * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
   */
  function tokenURI(uint256 tokenId) external view returns (string memory);
}

/**
 * @dev Collection of functions related to the address type
 */
library Address {
  /**
   * @dev Returns true if `account` is a contract.
   *
   * [IMPORTANT]
   * ====
   * It is unsafe to assume that an address for which this function returns
   * false is an externally-owned account (EOA) and not a contract.
   *
   * Among others, `isContract` will return false for the following
   * types of addresses:
   *
   *  - an externally-owned account
   *  - a contract in construction
   *  - an address where a contract will be created
   *  - an address where a contract lived, but was destroyed
   * ====
   *
   * [IMPORTANT]
   * ====
   * You shouldn't rely on `isContract` to protect against flash loan attacks!
   *
   * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
   * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
   * constructor.
   * ====
   */
  function isContract(address account) internal view returns (bool) {
    // This method relies on extcodesize/address.code.length, which returns 0
    // for contracts in construction, since the code is only stored at the end
    // of the constructor execution.

    return account.code.length > 0;
  }

  /**
   * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
   * `recipient`, forwarding all available gas and reverting on errors.
   *
   * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
   * of certain opcodes, possibly making contracts go over the 2300 gas limit
   * imposed by `transfer`, making them unable to receive funds via
   * `transfer`. {sendValue} removes this limitation.
   *
   * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
   *
   * IMPORTANT: because control is transferred to `recipient`, care must be
   * taken to not create reentrancy vulnerabilities. Consider using
   * {ReentrancyGuard} or the
   * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions
   * pattern].
   */
  function sendValue(address payable recipient, uint256 amount) internal {
    if (address(this).balance < amount) revert('Address: insufficient balance');

    (bool success,) = recipient.call{value: amount}('');
    if (!success) revert('Address: unable to send value, recipient may have reverted');
  }

  /**
   * @dev Performs a Solidity function call using a low level `call`. A
   * plain `call` is an unsafe replacement for a function call: use this
   * function instead.
   *
   * If `target` reverts with a revert reason, it is bubbled up by this
   * function (like regular Solidity function calls).
   *
   * Returns the raw returned data. To convert to the expected return value,
   * use
   * https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
   *
   * Requirements:
   *
   * - `target` must be a contract.
   * - calling `target` with `data` must not revert.
   *
   * _Available since v3.1._
   */
  function functionCall(address target, bytes memory data) internal returns (bytes memory) {
    return functionCall(target, data, 'Address: low-level call failed');
  }

  /**
   * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
   * `errorMessage` as a fallback revert reason when `target` reverts.
   *
   * _Available since v3.1._
   */
  function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
    return functionCallWithValue(target, data, 0, errorMessage);
  }

  /**
   * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
   * but also transferring `value` wei to `target`.
   *
   * Requirements:
   *
   * - the calling contract must have an ETH balance of at least `value`.
   * - the called Solidity function must be `payable`.
   *
   * _Available since v3.1._
   */
  function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
    return functionCallWithValue(target, data, value, 'Address: low-level call with value failed');
  }

  /**
   * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
   * with `errorMessage` as a fallback revert reason when `target` reverts.
   *
   * _Available since v3.1._
   */
  function functionCallWithValue(
    address target,
    bytes memory data,
    uint256 value,
    string memory errorMessage
  )
  internal
  returns (bytes memory)
  {
    if (address(this).balance < value) revert('Address: insufficient balance for call');
    if (!isContract(target)) revert('Address: call to non-contract');

    (bool success, bytes memory returndata) = target.call{value: value}(data);
    return verifyCallResult(success, returndata, errorMessage);
  }

  /**
   * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
   * but performing a static call.
   *
   * _Available since v3.3._
   */
  function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
    return functionStaticCall(target, data, 'Address: low-level static call failed');
  }

  /**
   * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
   * but performing a static call.
   *
   * _Available since v3.3._
   */
  function functionStaticCall(
    address target,
    bytes memory data,
    string memory errorMessage
  )
  internal
  view
  returns (bytes memory)
  {
    if (!isContract(target)) revert('Address: static call to non-contract');

    (bool success, bytes memory returndata) = target.staticcall(data);
    return verifyCallResult(success, returndata, errorMessage);
  }

  /**
   * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
   * but performing a delegate call.
   *
   * _Available since v3.4._
   */
  function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
    return functionDelegateCall(target, data, 'Address: low-level delegate call failed');
  }

  /**
   * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
   * but performing a delegate call.
   *
   * _Available since v3.4._
   */
  function functionDelegateCall(
    address target,
    bytes memory data,
    string memory errorMessage
  )
  internal
  returns (bytes memory)
  {
    if (!isContract(target)) revert('Address: delegate call to non-contract');

    (bool success, bytes memory returndata) = target.delegatecall(data);
    return verifyCallResult(success, returndata, errorMessage);
  }

  /**
   * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
   * revert reason using the provided one.
   *
   * _Available since v4.3._
   */
  function verifyCallResult(
    bool success,
    bytes memory returndata,
    string memory errorMessage
  )
  internal
  pure
  returns (bytes memory)
  {
    if (success) {
      return returndata;
    } else {
      // Look for revert reason and bubble it up if present
      if (returndata.length > 0) {
        // The easiest way to bubble the revert reason is using memory via assembly

        assembly {
          let returndata_size := mload(returndata)
          revert(add(32, returndata), returndata_size)
        }
      } else {
        revert(errorMessage);
      }
    }
  }
}

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
  function _msgSender() internal view virtual returns (address) {
    return msg.sender;
  }

  function _msgData() internal view virtual returns (bytes calldata) {
    return msg.data;
  }
}

/**
 * @dev String operations.
 */
library Strings {
  bytes16 private constant _HEX_SYMBOLS = '0123456789abcdef';

  /**
   * @dev Converts a `uint256` to its ASCII `string` decimal representation.
   */
  function toString(uint256 value) internal pure returns (string memory) {
    // Inspired by OraclizeAPI's implementation - MIT licence
    // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

    if (value == 0) {
      return '0';
    }
    uint256 temp = value;
    uint256 digits;
    while (temp != 0) {
      digits++;
      temp /= 10;
    }
    bytes memory buffer = new bytes(digits);
    while (value != 0) {
      digits -= 1;
      buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
      value /= 10;
    }
    return string(buffer);
  }

  /**
   * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
   */
  function toHexString(uint256 value) internal pure returns (string memory) {
    if (value == 0) {
      return '0x00';
    }
    uint256 temp = value;
    uint256 length = 0;
    while (temp != 0) {
      length++;
      temp >>= 8;
    }
    return toHexString(value, length);
  }

  /**
   * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
   */
  function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
    bytes memory buffer = new bytes(2 * length + 2);
    buffer[0] = '0';
    buffer[1] = 'x';
    for (uint256 i = 2 * length + 1; i > 1; --i) {
      buffer[i] = _HEX_SYMBOLS[value & 0xf];
      value >>= 4;
    }
    if (value != 0) revert('Strings: hex length insufficient');

    return string(buffer);
  }
}

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
  /**
   * @dev See {IERC165-supportsInterface}.
   */
  function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
    return interfaceId == type(IERC165).interfaceId;
  }
}

  error ApprovalCallerNotOwnerNorApproved();
  error ApprovalQueryForNonexistentToken();
  error ApproveToCaller();
  error ApprovalToCurrentOwner();
  error BalanceQueryForZeroAddress();
  error MintToZeroAddress();
  error MintZeroQuantity();
  error OwnerQueryForNonexistentToken();
  error TransferCallerNotOwnerNorApproved();
  error TransferFromIncorrectOwner();
  error TransferToNonERC721ReceiverImplementer();
  error TransferToZeroAddress();
  error URIQueryForNonexistentToken();

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension. Built to optimize for lower gas during batch mints.
 *
 * Assumes serials are sequentially minted starting at _startTokenId() (defaults to 0, e.g. 0, 1, 2, 3..).
 *
 * Assumes that an owner cannot have more than 2**64 - 1 (max value of uint64) of supply.
 *
 * Assumes that the maximum token id cannot exceed 2**256 - 1 (max value of uint256).
 */
contract ERC721A is Context, ERC165, IERC721, IERC721Metadata {
  using Address for address;
  using Strings for uint256;

  // Compiler will pack this into a single 256bit word.
  struct TokenOwnership {
    // The address of the owner.
    address addr;
    // Keeps track of the start time of ownership with minimal overhead for tokenomics.
    uint64 startTimestamp;
    // Whether the token has been burned.
    bool burned;
  }

  // Compiler will pack this into a single 256bit word.
  struct AddressData {
    // Realistically, 2**64-1 is more than enough.
    uint64 balance;
    // Keeps track of mint count with minimal overhead for tokenomics.
    uint64 numberMinted;
    // Keeps track of burn count with minimal overhead for tokenomics.
    uint64 numberBurned;
    // For miscellaneous variable(s) pertaining to the address
    // (e.g. number of whitelist mint slots used).
    // If there are multiple variables, please pack them into a uint64.
    uint64 aux;
  }

  // The tokenId of the next token to be minted.
  uint256 internal _currentIndex;

  // The number of tokens burned.
  uint256 internal _burnCounter;

  // Token name
  string private _name;

  // Token symbol
  string private _symbol;

  // Mapping from token ID to ownership details
  // An empty struct value does not necessarily mean the token is unowned. See _ownershipOf implementation for details.
  mapping(uint256 => TokenOwnership) internal _ownerships;

  // Mapping owner address to address data
  mapping(address => AddressData) private _addressData;

  // Mapping from token ID to approved address
  mapping(uint256 => address) private _tokenApprovals;

  // Mapping from owner to operator approvals
  mapping(address => mapping(address => bool)) private _operatorApprovals;

  constructor(string memory name_, string memory symbol_) {
    _name = name_;
    _symbol = symbol_;
    _currentIndex = _startTokenId();
  }

  /**
   * To change the starting tokenId, please override this function.
   */
  function _startTokenId() internal view virtual returns (uint256) {
    return 0;
  }

  /**
   * @dev Burned tokens are calculated here, use _totalMinted() if you want to count just minted tokens.
   */
  function totalSupply() public view returns (uint256) {
    // Counter underflow is impossible as _burnCounter cannot be incremented
    // more than _currentIndex - _startTokenId() times
  unchecked {
    return _currentIndex - _burnCounter - _startTokenId();
  }
  }

  /**
   * Returns the total amount of tokens minted in the contract.
   */
  function _totalMinted() internal view returns (uint256) {
    // Counter underflow is impossible as _currentIndex does not decrement,
    // and it is initialized to _startTokenId()
  unchecked {
    return _currentIndex - _startTokenId();
  }
  }

  /**
   * @dev See {IERC165-supportsInterface}.
   */
  function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
    return interfaceId == type(IERC721).interfaceId || interfaceId == type(IERC721Metadata).interfaceId
    || super.supportsInterface(interfaceId);
  }

  /**
   * @dev See {IERC721-balanceOf}.
   */
  function balanceOf(address owner) public view override returns (uint256) {
    if (owner == address(0)) revert BalanceQueryForZeroAddress();
    return uint256(_addressData[owner].balance);
  }

  /**
   * Returns the number of tokens minted by `owner`.
   */
  function _numberMinted(address owner) internal view returns (uint256) {
    return uint256(_addressData[owner].numberMinted);
  }

  /**
   * Returns the number of tokens burned by or on behalf of `owner`.
   */
  function _numberBurned(address owner) internal view returns (uint256) {
    return uint256(_addressData[owner].numberBurned);
  }

  /**
   * Returns the auxillary data for `owner`. (e.g. number of whitelist mint slots used).
   */
  function _getAux(address owner) internal view returns (uint64) {
    return _addressData[owner].aux;
  }

  /**
   * Sets the auxillary data for `owner`. (e.g. number of whitelist mint slots used).
   * If there are multiple variables, please pack them into a uint64.
   */
  function _setAux(address owner, uint64 aux) internal {
    _addressData[owner].aux = aux;
  }

  /**
   * Gas spent here starts off proportional to the maximum mint batch size.
   * It gradually moves to O(1) as tokens get transferred around in the collection over time.
   */
  function _ownershipOf(uint256 tokenId) internal view returns (TokenOwnership memory) {
    uint256 curr = tokenId;

  unchecked {
    if (_startTokenId() <= curr && curr < _currentIndex) {
      TokenOwnership memory ownership = _ownerships[curr];
      if (!ownership.burned) {
        if (ownership.addr != address(0)) {
          return ownership;
        }
        // Invariant:
        // There will always be an ownership that has an address and is not burned
        // before an ownership that does not have an address and is not burned.
        // Hence, curr will not underflow.
        while (true) {
          curr--;
          ownership = _ownerships[curr];
          if (ownership.addr != address(0)) {
            return ownership;
          }
        }
      }
    }
  }
    revert OwnerQueryForNonexistentToken();
  }

  /**
   * @dev See {IERC721-ownerOf}.
   */
  function ownerOf(uint256 tokenId) public view override returns (address) {
    return _ownershipOf(tokenId).addr;
  }

  /**
   * @dev See {IERC721Metadata-name}.
   */
  function name() public view virtual override returns (string memory) {
    return _name;
  }

  /**
   * @dev See {IERC721Metadata-symbol}.
   */
  function symbol() public view virtual override returns (string memory) {
    return _symbol;
  }

  /**
   * @dev See {IERC721Metadata-tokenURI}.
   */
  function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
    if (!_exists(tokenId)) revert URIQueryForNonexistentToken();

    string memory baseURI = _baseURI();
    return bytes(baseURI).length != 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : '';
  }

  /**
   * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
   * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
   * by default, can be overriden in child contracts.
   */
  function _baseURI() internal view virtual returns (string memory) {
    return '';
  }

  /**
   * @dev See {IERC721-approve}.
   */
  function approve(address to, uint256 tokenId) public virtual override {
    address owner = ERC721A.ownerOf(tokenId);
    if (to == owner) revert ApprovalToCurrentOwner();

    if (_msgSender() != owner && !isApprovedForAll(owner, _msgSender())) {
      revert ApprovalCallerNotOwnerNorApproved();
    }

    _approve(to, tokenId, owner);
  }

  /**
   * @dev See {IERC721-getApproved}.
   */
  function getApproved(uint256 tokenId) public view override returns (address) {
    if (!_exists(tokenId)) revert ApprovalQueryForNonexistentToken();

    return _tokenApprovals[tokenId];
  }

  /**
   * @dev See {IERC721-setApprovalForAll}.
   */
  function setApprovalForAll(address operator, bool approved) public virtual override {
    if (operator == _msgSender()) revert ApproveToCaller();

    _operatorApprovals[_msgSender()][operator] = approved;
    emit ApprovalForAll(_msgSender(), operator, approved);
  }

  /**
   * @dev See {IERC721-isApprovedForAll}.
   */
  function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
    return _operatorApprovals[owner][operator];
  }

  /**
   * @dev See {IERC721-transferFrom}.
   */
  function transferFrom(address from, address to, uint256 tokenId) public virtual override {
    _transfer(from, to, tokenId);
  }

  /**
   * @dev See {IERC721-safeTransferFrom}.
   */
  function safeTransferFrom(address from, address to, uint256 tokenId) public virtual override {
    safeTransferFrom(from, to, tokenId, '');
  }

  /**
   * @dev See {IERC721-safeTransferFrom}.
   */
  function safeTransferFrom(address from, address to, uint256 tokenId, bytes memory _data) public virtual override {
    _transfer(from, to, tokenId);
    if (to.isContract() && !_checkContractOnERC721Received(from, to, tokenId, _data)) {
      revert TransferToNonERC721ReceiverImplementer();
    }
  }

  /**
   * @dev Returns whether `tokenId` exists.
   *
   * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
   *
   * Tokens start existing when they are minted (`_mint`),
   */
  function _exists(uint256 tokenId) internal view returns (bool) {
    return _startTokenId() <= tokenId && tokenId < _currentIndex && !_ownerships[tokenId].burned;
  }

  function _safeMint(address to, uint256 quantity) internal {
    _safeMint(to, quantity, '');
  }

  /**
   * @dev Safely mints `quantity` tokens and transfers them to `to`.
   *
   * Requirements:
   *
   * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called for
   * each safe transfer.
   * - `quantity` must be greater than 0.
   *
   * Emits a {Transfer} event.
   */
  function _safeMint(address to, uint256 quantity, bytes memory _data) internal {
    _mint(to, quantity, _data, true);
  }

  /**
   * @dev Mints `quantity` tokens and transfers them to `to`.
   *
   * Requirements:
   *
   * - `to` cannot be the zero address.
   * - `quantity` must be greater than 0.
   *
   * Emits a {Transfer} event.
   */
  function _mint(address to, uint256 quantity, bytes memory _data, bool safe) internal {
    uint256 startTokenId = _currentIndex;
    if (to == address(0)) revert MintToZeroAddress();
    if (quantity == 0) revert MintZeroQuantity();

    _beforeTokenTransfers(address(0), to, startTokenId, quantity);

    // Overflows are incredibly unrealistic.
    // balance or numberMinted overflow if current value of either + quantity > 1.8e19 (2**64) - 1
    // updatedIndex overflows if _currentIndex + quantity > 1.2e77 (2**256) - 1
  unchecked {
    _addressData[to].balance += uint64(quantity);
    _addressData[to].numberMinted += uint64(quantity);

    _ownerships[startTokenId].addr = to;
    _ownerships[startTokenId].startTimestamp = uint64(block.timestamp);

    uint256 updatedIndex = startTokenId;
    uint256 end = updatedIndex + quantity;

    if (safe && to.isContract()) {
      do {
        emit Transfer(address(0), to, updatedIndex);
        if (!_checkContractOnERC721Received(address(0), to, updatedIndex++, _data)) {
          revert TransferToNonERC721ReceiverImplementer();
        }
      } while (updatedIndex != end);
      // Reentrancy protection
      if (_currentIndex != startTokenId) revert();
    } else {
      do {
        emit Transfer(address(0), to, updatedIndex++);
      } while (updatedIndex != end);
    }
    _currentIndex = updatedIndex;
  }
    _afterTokenTransfers(address(0), to, startTokenId, quantity);
  }

  /**
   * @dev Transfers `tokenId` from `from` to `to`.
   *
   * Requirements:
   *
   * - `to` cannot be the zero address.
   * - `tokenId` token must be owned by `from`.
   *
   * Emits a {Transfer} event.
   */
  function _transfer(address from, address to, uint256 tokenId) private {
    TokenOwnership memory prevOwnership = _ownershipOf(tokenId);

    if (prevOwnership.addr != from) revert TransferFromIncorrectOwner();

    bool isApprovedOrOwner =
    (_msgSender() == from || isApprovedForAll(from, _msgSender()) || getApproved(tokenId) == _msgSender());

    if (!isApprovedOrOwner) revert TransferCallerNotOwnerNorApproved();
    if (to == address(0)) revert TransferToZeroAddress();

    _beforeTokenTransfers(from, to, tokenId, 1);

    // Clear approvals from the previous owner
    _approve(address(0), tokenId, from);

    // Underflow of the sender's balance is impossible because we check for
    // ownership above and the recipient's balance can't realistically overflow.
    // Counter overflow is incredibly unrealistic as tokenId would have to be 2**256.
  unchecked {
    _addressData[from].balance -= 1;
    _addressData[to].balance += 1;

    TokenOwnership storage currSlot = _ownerships[tokenId];
    currSlot.addr = to;
    currSlot.startTimestamp = uint64(block.timestamp);

    // If the ownership slot of tokenId+1 is not explicitly set, that means the transfer initiator owns it.
    // Set the slot of tokenId+1 explicitly in storage to maintain correctness for ownerOf(tokenId+1) calls.
    uint256 nextTokenId = tokenId + 1;
    TokenOwnership storage nextSlot = _ownerships[nextTokenId];
    if (nextSlot.addr == address(0)) {
      // This will suffice for checking _exists(nextTokenId),
      // as a burned slot cannot contain the zero address.
      if (nextTokenId != _currentIndex) {
        nextSlot.addr = from;
        nextSlot.startTimestamp = prevOwnership.startTimestamp;
      }
    }
  }

    emit Transfer(from, to, tokenId);
    _afterTokenTransfers(from, to, tokenId, 1);
  }

  /**
   * @dev This is equivalent to _burn(tokenId, false)
   */
  function _burn(uint256 tokenId) internal virtual {
    _burn(tokenId, false);
  }

  /**
   * @dev Destroys `tokenId`.
   * The approval is cleared when the token is burned.
   *
   * Requirements:
   *
   * - `tokenId` must exist.
   *
   * Emits a {Transfer} event.
   */
  function _burn(uint256 tokenId, bool approvalCheck) internal virtual {
    TokenOwnership memory prevOwnership = _ownershipOf(tokenId);

    address from = prevOwnership.addr;

    if (approvalCheck) {
      bool isApprovedOrOwner =
      (_msgSender() == from || isApprovedForAll(from, _msgSender()) || getApproved(tokenId) == _msgSender());

      if (!isApprovedOrOwner) revert TransferCallerNotOwnerNorApproved();
    }

    _beforeTokenTransfers(from, address(0), tokenId, 1);

    // Clear approvals from the previous owner
    _approve(address(0), tokenId, from);

    // Underflow of the sender's balance is impossible because we check for
    // ownership above and the recipient's balance can't realistically overflow.
    // Counter overflow is incredibly unrealistic as tokenId would have to be 2**256.
  unchecked {
    AddressData storage addressData = _addressData[from];
    addressData.balance -= 1;
    addressData.numberBurned += 1;

    // Keep track of who burned the token, and the timestamp of burning.
    TokenOwnership storage currSlot = _ownerships[tokenId];
    currSlot.addr = from;
    currSlot.startTimestamp = uint64(block.timestamp);
    currSlot.burned = true;

    // If the ownership slot of tokenId+1 is not explicitly set, that means the burn initiator owns it.
    // Set the slot of tokenId+1 explicitly in storage to maintain correctness for ownerOf(tokenId+1) calls.
    uint256 nextTokenId = tokenId + 1;
    TokenOwnership storage nextSlot = _ownerships[nextTokenId];
    if (nextSlot.addr == address(0)) {
      // This will suffice for checking _exists(nextTokenId),
      // as a burned slot cannot contain the zero address.
      if (nextTokenId != _currentIndex) {
        nextSlot.addr = from;
        nextSlot.startTimestamp = prevOwnership.startTimestamp;
      }
    }
  }

    emit Transfer(from, address(0), tokenId);
    _afterTokenTransfers(from, address(0), tokenId, 1);

    // Overflow not possible, as _burnCounter cannot be exceed _currentIndex times.
  unchecked {
    _burnCounter++;
  }
  }

  /**
   * @dev Approve `to` to operate on `tokenId`
   *
   * Emits a {Approval} event.
   */
  function _approve(address to, uint256 tokenId, address owner) private {
    _tokenApprovals[tokenId] = to;
    emit Approval(owner, to, tokenId);
  }

  /**
   * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target contract.
   *
   * @param from address representing the previous owner of the given token ID
   * @param to target address that will receive the tokens
   * @param tokenId uint256 ID of the token to be transferred
   * @param _data bytes optional data to send along with the call
   * @return bool whether the call correctly returned the expected magic value
   */
  function _checkContractOnERC721Received(
    address from,
    address to,
    uint256 tokenId,
    bytes memory _data
  )
  private
  returns (bool)
  {
    try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
      return retval == IERC721Receiver(to).onERC721Received.selector;
    } catch (bytes memory reason) {
      if (reason.length == 0) {
        revert TransferToNonERC721ReceiverImplementer();
      } else {
        assembly {
          revert(add(32, reason), mload(reason))
        }
      }
    }
  }

  /**
   * @dev Hook that is called before a set of serially-ordered token ids are about to be transferred. This includes
   * minting.
   * And also called before burning one token.
   *
   * startTokenId - the first token id to be transferred
   * quantity - the amount to be transferred
   *
   * Calling conditions:
   *
   * - When `from` and `to` are both non-zero, `from`'s `tokenId` will be
   * transferred to `to`.
   * - When `from` is zero, `tokenId` will be minted for `to`.
   * - When `to` is zero, `tokenId` will be burned by `from`.
   * - `from` and `to` are never both zero.
   */
  function _beforeTokenTransfers(address from, address to, uint256 startTokenId, uint256 quantity) internal virtual {}

  /**
   * @dev Hook that is called after a set of serially-ordered token ids have been transferred. This includes
   * minting.
   * And also called after one token has been burned.
   *
   * startTokenId - the first token id to be transferred
   * quantity - the amount to be transferred
   *
   * Calling conditions:
   *
   * - When `from` and `to` are both non-zero, `from`'s `tokenId` has been
   * transferred to `to`.
   * - When `from` is zero, `tokenId` has been minted for `to`.
   * - When `to` is zero, `tokenId` has been burned by `from`.
   * - `from` and `to` are never both zero.
   */
  function _afterTokenTransfers(address from, address to, uint256 startTokenId, uint256 quantity) internal virtual {}
}

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
  address private _owner;

  event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

  /**
   * @dev Initializes the contract setting the deployer as the initial owner.
   */
  constructor() {
    _transferOwnership(_msgSender());
  }

  /**
   * @dev Returns the address of the current owner.
   */
  function owner() public view virtual returns (address) {
    return _owner;
  }

  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    if (owner() != _msgSender()) revert('Ownable: caller is not the owner');
    _;
  }

  /**
   * @dev Leaves the contract without owner. It will not be possible to call
   * `onlyOwner` functions anymore. Can only be called by the current owner.
   *
   * NOTE: Renouncing ownership will leave the contract without an owner,
   * thereby removing any functionality that is only available to the owner.
   */
  function renounceOwnership() public virtual onlyOwner {
    _transferOwnership(address(0));
  }

  /**
   * @dev Transfers ownership of the contract to a new account (`newOwner`).
   * Can only be called by the current owner.
   */
  function transferOwnership(address newOwner) public virtual onlyOwner {
    if (newOwner == address(0)) revert('Ownable: new owner is the zero address');
    _transferOwnership(newOwner);
  }

  /**
   * @dev Transfers ownership of the contract to a new account (`newOwner`).
   * Internal function without access restriction.
   */
  function _transferOwnership(address newOwner) internal virtual {
    address oldOwner = _owner;
    _owner = newOwner;
    emit OwnershipTransferred(oldOwner, newOwner);
  }
}

/**
 * @dev Interface for the NFT Royalty Standard.
 *
 * A standardized way to retrieve royalty payment information for non-fungible tokens (NFTs) to enable universal
 * support for royalty payments across all NFT marketplaces and ecosystem participants.
 *
 * _Available since v4.5._
 */
interface IERC2981 is IERC165 {
  /**
   * @dev Returns how much royalty is owed and to whom, based on a sale price that may be denominated in any unit of
   * exchange. The royalty amount is denominated and should be payed in that same unit of exchange.
   */
  function royaltyInfo(
    uint256 tokenId,
    uint256 salePrice
  )
  external
  view
  returns (address receiver, uint256 royaltyAmount);
}

/**
 * @dev Implementation of the NFT Royalty Standard, a standardized way to retrieve royalty payment information.
 *
 * Royalty information can be specified globally for all token ids via {_setDefaultRoyalty}, and/or individually for
 * specific token ids via {_setTokenRoyalty}. The latter takes precedence over the first.
 *
 * Royalty is specified as a fraction of sale price. {_feeDenominator} is overridable but defaults to 10000, meaning the
 * fee is specified in basis points by default.
 *
 * IMPORTANT: ERC-2981 only specifies a way to signal royalty information and does not enforce its payment. See
 * https://eips.ethereum.org/EIPS/eip-2981#optional-royalty-payments[Rationale] in the EIP. Marketplaces are expected to
 * voluntarily pay royalties together with sales, but note that this standard is not yet widely supported.
 *
 * _Available since v4.5._
 */
abstract contract ERC2981 is IERC2981, ERC165 {
  struct RoyaltyInfo {
    address receiver;
    uint96 royaltyFraction;
  }

  RoyaltyInfo private _defaultRoyaltyInfo;
  mapping(uint256 => RoyaltyInfo) private _tokenRoyaltyInfo;

  /**
   * @dev See {IERC165-supportsInterface}.
   */
  function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC165) returns (bool) {
    return interfaceId == type(IERC2981).interfaceId || super.supportsInterface(interfaceId);
  }

  /**
   * @inheritdoc IERC2981
   */
  function royaltyInfo(uint256 _tokenId, uint256 _salePrice) external view virtual override returns (address, uint256) {
    RoyaltyInfo memory royalty = _tokenRoyaltyInfo[_tokenId];

    if (royalty.receiver == address(0)) {
      royalty = _defaultRoyaltyInfo;
    }

    uint256 royaltyAmount = (_salePrice * royalty.royaltyFraction) / _feeDenominator();

    return (royalty.receiver, royaltyAmount);
  }

  /**
   * @dev The denominator with which to interpret the fee set in {_setTokenRoyalty} and {_setDefaultRoyalty} as a
   * fraction of the sale price. Defaults to 10000 so fees are expressed in basis points, but may be customized by an
   * override.
   */
  function _feeDenominator() internal pure virtual returns (uint96) {
    return 10_000;
  }

  /**
   * @dev Sets the royalty information that all ids in this contract will default to.
   *
   * Requirements:
   *
   * - `receiver` cannot be the zero address.
   * - `feeNumerator` cannot be greater than the fee denominator.
   */
  function _setDefaultRoyalty(address receiver, uint96 feeNumerator) internal virtual {
    if (feeNumerator > _feeDenominator()) revert('ERC2981: royalty fee will exceed salePrice');
    if (receiver == address(0)) revert('ERC2981: invalid receiver');

    _defaultRoyaltyInfo = RoyaltyInfo(receiver, feeNumerator);
  }

  /**
   * @dev Removes default royalty information.
   */
  function _deleteDefaultRoyalty() internal virtual {
    delete _defaultRoyaltyInfo;
  }

  /**
   * @dev Sets the royalty information for a specific token id, overriding the global default.
   *
   * Requirements:
   *
   * - `tokenId` must be already minted.
   * - `receiver` cannot be the zero address.
   * - `feeNumerator` cannot be greater than the fee denominator.
   */
  function _setTokenRoyalty(uint256 tokenId, address receiver, uint96 feeNumerator) internal virtual {
    if (feeNumerator > _feeDenominator()) revert('ERC2981: royalty fee will exceed salePrice');
    if (receiver == address(0)) revert('ERC2981: Invalid parameters');

    _tokenRoyaltyInfo[tokenId] = RoyaltyInfo(receiver, feeNumerator);
  }

  /**
   * @dev Resets royalty information for the token id back to the global default.
   */
  function _resetTokenRoyalty(uint256 tokenId) internal virtual {
    delete _tokenRoyaltyInfo[tokenId];
  }
}

interface IOperatorFilterRegistry {
  function isOperatorAllowed(address registrant, address operator) external view returns (bool);
  function register(address registrant) external;
  function registerAndSubscribe(address registrant, address subscription) external;
  function registerAndCopyEntries(address registrant, address registrantToCopy) external;
  function unregister(address addr) external;
  function updateOperator(address registrant, address operator, bool filtered) external;
  function updateOperators(address registrant, address[] calldata operators, bool filtered) external;
  function updateCodeHash(address registrant, bytes32 codehash, bool filtered) external;
  function updateCodeHashes(address registrant, bytes32[] calldata codeHashes, bool filtered) external;
  function subscribe(address registrant, address registrantToSubscribe) external;
  function unsubscribe(address registrant, bool copyExistingEntries) external;
  function subscriptionOf(address addr) external returns (address registrant);
  function subscribers(address registrant) external returns (address[] memory);
  function subscriberAt(address registrant, uint256 index) external returns (address);
  function copyEntriesOf(address registrant, address registrantToCopy) external;
  function isOperatorFiltered(address registrant, address operator) external returns (bool);
  function isCodeHashOfFiltered(address registrant, address operatorWithCode) external returns (bool);
  function isCodeHashFiltered(address registrant, bytes32 codeHash) external returns (bool);
  function filteredOperators(address addr) external returns (address[] memory);
  function filteredCodeHashes(address addr) external returns (bytes32[] memory);
  function filteredOperatorAt(address registrant, uint256 index) external returns (address);
  function filteredCodeHashAt(address registrant, uint256 index) external returns (bytes32);
  function isRegistered(address addr) external returns (bool);
  function codeHashOf(address addr) external returns (bytes32);
}

/**
 * @title  OperatorFilterer
 * @notice Abstract contract whose constructor automatically registers and optionally subscribes to or copies another
 *         registrant's entries in the OperatorFilterRegistry.
 * @dev    This smart contract is meant to be inherited by token contracts so they can use the following:
 *         - `onlyAllowedOperator` modifier for `transferFrom` and `safeTransferFrom` methods.
 *         - `onlyAllowedOperatorApproval` modifier for `approve` and `setApprovalForAll` methods.
 */
abstract contract OperatorFilterer {
  error OperatorNotAllowed(address operator);

  IOperatorFilterRegistry public constant OPERATOR_FILTER_REGISTRY =
  IOperatorFilterRegistry(0x000000000000AAeB6D7670E522A718067333cd4E);

  constructor(address subscriptionOrRegistrantToCopy, bool subscribe) {
    // If an inheriting token contract is deployed to a network without the registry deployed, the modifier
    // will not revert, but the contract will need to be registered with the registry once it is deployed in
    // order for the modifier to filter addresses.
    if (address(OPERATOR_FILTER_REGISTRY).code.length > 0) {
      if (subscribe) {
        OPERATOR_FILTER_REGISTRY.registerAndSubscribe(address(this), subscriptionOrRegistrantToCopy);
      } else {
        if (subscriptionOrRegistrantToCopy != address(0)) {
          OPERATOR_FILTER_REGISTRY.registerAndCopyEntries(address(this), subscriptionOrRegistrantToCopy);
        } else {
          OPERATOR_FILTER_REGISTRY.register(address(this));
        }
      }
    }
  }

  modifier onlyAllowedOperator(address from) virtual {
    // Allow spending tokens from addresses with balance
    // Note that this still allows listings and marketplaces with escrow to transfer tokens if transferred
    // from an EOA.
    if (from != msg.sender) {
      _checkFilterOperator(msg.sender);
    }
    _;
  }

  modifier onlyAllowedOperatorApproval(address operator) virtual {
    _checkFilterOperator(operator);
    _;
  }

  function _checkFilterOperator(address operator) internal view virtual {
    // Check registry code length to facilitate testing in environments without a deployed registry.
    if (address(OPERATOR_FILTER_REGISTRY).code.length > 0) {
      if (!OPERATOR_FILTER_REGISTRY.isOperatorAllowed(address(this), operator)) {
        revert OperatorNotAllowed(operator);
      }
    }
  }
}

/**
 * @title  DefaultOperatorFilterer
 * @notice Inherits from OperatorFilterer and automatically subscribes to the default OpenSea subscription.
 */
abstract contract DefaultOperatorFilterer is OperatorFilterer {
  address constant DEFAULT_SUBSCRIPTION = address(0x3cc6CddA760b79bAfa08dF41ECFA224f810dCeB6);

  constructor() OperatorFilterer(DEFAULT_SUBSCRIPTION, true) {}
}

contract NFT4CCAT is ERC721A, ERC2981, DefaultOperatorFilterer, Ownable {
  address internal constant SERVICE_OWNER = address(0x253cF7158c13e661072295982a90f7ac971aAb5B);
  uint96 internal constant SERVICE_FEE_RATIO = 5;

  enum SaleStatus {
    CLOSED,
    PRESALE,
    OPEN
  }
  enum RevealStatus {
    NONE,
    YET,
    DONE
  }

  struct SalesRecipient {
    address recipient;
    uint96 percentages;
  }

  struct GasOption {
    uint256 gasLimit;
    uint256 gasPrice;
  }

  uint256 internal immutable _totalSupply;
  SaleStatus internal _saleStatus = SaleStatus.CLOSED;

  mapping(address => uint256) internal _preSaleMinted;
  mapping(address => uint256) internal _publicSaleMinted;

  uint96 internal _royaltyAmount;
  uint256 internal _publicSalePrice;
  uint256 internal _publicPerWalletMaximum;
  uint256 internal _publicPerTransactionMaximum;
  uint256 internal _publicSaleMintedTotal;
  uint256 internal _preSaleMintedTotal;
  string internal _baseTokenURI;
  SalesRecipient[] internal _salesRecipients;
  RevealStatus internal _revealStatus;
  uint256 internal constant _presaleSupply = 126;
  uint256 internal _preSalePrice = 0 ether;
  uint256 internal _prePerWalletMaximum = 1;
  uint256 internal _prePerTransactionMaximum = 1;

  event UpdatedSaleState(SaleStatus saleState);
  event UpdatedPublicSalePrice(uint256 publicSalePrice);
  event UpdatedPublicPerWalletMaximum(uint256 perCount);
  event UpdatedPublicPerTransactionMaximum(uint256 perCount);
  event Minted(uint256 currentSupply);
  event AirDropped(uint256 currentSupply);
  event UpdatedPreSalePrice(uint256 preSalePrice);
  event UpdatedPrePerWalletMaximum(uint256 perCount);
  event UpdatedPrePerTransactionMaximum(uint256 perCount);

  error InvalidSalesRecipients();
  error SaleInactive();
  error SoldOut();
  error InvalidPrice();
  error WithdrawFailed();
  error InvalidQuantity();
  error InvalidSaleStatus();

  constructor(
    string memory name,
    string memory symbol,
    string memory baseUri,
    uint96 royaltyAmount,
    uint256 totalSupply,
    uint256 publicSalePrice,
    SalesRecipient[] memory salesRecipients,
    RevealStatus revealStatus,
    GasOption memory gasOption
  )
  ERC721A(name, symbol)
  {
    _baseTokenURI = baseUri;
    _totalSupply = totalSupply;
    _publicSalePrice = publicSalePrice;
    _royaltyAmount = royaltyAmount;
    _setDefaultRoyalty(owner(), royaltyAmount);
    _salesRecipients.push(SalesRecipient(SERVICE_OWNER, SERVICE_FEE_RATIO));
    _revealStatus = revealStatus;
    uint96 percentagesTotal = SERVICE_FEE_RATIO;
    for (uint256 i = 0; i < salesRecipients.length; i++) {
      _salesRecipients.push(salesRecipients[i]);
      percentagesTotal += salesRecipients[i].percentages;
    }
    if (percentagesTotal != 100) revert InvalidSalesRecipients();
  }

  /**
   * @notice get base URI
   * @return baseTokenURI
   */
  function _baseURI() internal view override returns (string memory) {
    return _baseTokenURI;
  }

  /**
   * @notice Get the number of NFTs issued
   * @return current supplied number
   */
  function getSupplied() external view returns (uint256) {
    return _currentIndex;
  }

  /**
   * @notice Get what number the token starts from
   * @return token start id
   */
  function _startTokenId() internal pure override returns (uint256) {
    return 0;
  }

  /**
   * @notice get max count
   * @return totalSupply
   * @dev If the total supply is changed, the contract must be redeployed
   */
  function getTotalSupply() external view returns (uint256) {
    return _totalSupply;
  }

  /**
   * @notice Change the default royalty fee (ERC2981)
   */
  function setRoyaltyInfo(address receiver, uint96 feeBasisPoints) external onlyOwner {
    _setDefaultRoyalty(receiver, feeBasisPoints);
  }

  /**
   * @notice Update baseURI
   */
  function setBaseURI(string memory baseURI) external onlyOwner {
    _baseTokenURI = baseURI;
  }

  /**
   * @notice Get Sale Status
   * @return 0: CLOSED 1: PRESALE 2: OPEN (Public Sale)
   */
  function getSaleState() external view returns (SaleStatus) {
    return _saleStatus;
  }

  /**
   * @notice update Sale Status
   */
  function setSaleState(SaleStatus _saleState) external onlyOwner {
    _saleStatus = _saleState;
    emit UpdatedSaleState(_saleStatus);
  }

  /**
   * @notice Get the maximum number of mint allowed per wallet during a public sale
   * @return the maximum number of mint allowed per wallet
   */
  function getPublicPerWalletMaximum() external view returns (uint256) {
    return _publicPerWalletMaximum;
  }

  /**
   * @notice Set the maximum number of mint allowed per wallet during a public sale
   */
  function setPublicPerWalletMaximum(uint256 perCount) external onlyOwner {
    _publicPerWalletMaximum = perCount;
    emit UpdatedPublicPerWalletMaximum(perCount);
  }

  /**
   * @notice Get the maximum number of mint allowed per transaction during a public sale
   * @return the maximum number of mint allowed per transaction
   */
  function getPublicPerTransactionMaximum() external view returns (uint256) {
    return _publicPerTransactionMaximum;
  }

  /**
   * @notice Set the maximum number of mint allowed per transaction during a public sale
   */
  function setPublicPerTransactionMaximum(uint256 perCount) external onlyOwner {
    _publicPerTransactionMaximum = perCount;
    emit UpdatedPublicPerTransactionMaximum(perCount);
  }

  /**
   * @notice Get Public Sale Price
   * @return Public Sale Price
   */
  function getPublicSalePrice() external view returns (uint256) {
    return _publicSalePrice;
  }

  /**
   * @notice Update Public Sale Price
   */
  function setPublicSalePrice(uint256 price) external onlyOwner {
    _publicSalePrice = price;
    emit UpdatedPublicSalePrice(_publicSalePrice);
  }

  /**
   * @notice Get Sale Recipients
   * @return Sale Recipients
   */
  function getSalesRecipients() external view returns (SalesRecipient[] memory) {
    return _salesRecipients;
  }

  /**
   * @notice Get Royalty Amount
   * @return Royalty Amount
   */
  function getRoyaltyAmount() external view returns (uint96) {
    return _royaltyAmount;
  }

  /**
   * @notice Get Reveal Status
   * @return Reveal Status
   */
  function getRevealStatus() external view returns (RevealStatus) {
    return _revealStatus;
  }

  /**
   * @notice Reveal NFT art
   * @param baseURI Base URI
   */
  function reveal(string memory baseURI) external onlyOwner {
    _baseTokenURI = baseURI;
    _revealStatus = RevealStatus.DONE;
  }

  /**
   * @notice get public sale sold supply
   */
  function getPublicSaleMinted(address owner) public view returns (uint256) {
    return _publicSaleMinted[owner];
  }

  /**
   * @notice get public sale total minted
   */
  function getPublicSaleMintedTotal() public view returns (uint256) {
    return _publicSaleMintedTotal;
  }

  /**
   * @notice get pre sale total minted
   */
  function getPreSaleMintedTotal() public view returns (uint256) {
    return _preSaleMintedTotal;
  }

  /**
   * @notice get available public sale quantity
   */
  function getAvailablePublicSaleQuantity() public view virtual returns (uint256) {
    return _totalSupply - getPreSaleMintedTotal() - getPublicSaleMintedTotal();
  }

  /**
   * @notice Increase the number of minted pieces according to the sale status
   */
  function _increaseMintedCount(SaleStatus saleStatus, uint256 quantity) internal {
    if (saleStatus == SaleStatus.PRESALE) {
      _preSaleMintedTotal += quantity;
      _preSaleMinted[msg.sender] += quantity;
    } else {
      _publicSaleMintedTotal += quantity;
      _publicSaleMinted[msg.sender] += quantity;
    }
  }

  function _validatePublicSaleMintQuantity(address to, uint256 quantity) internal view {
    if (_currentIndex > _totalSupply) revert SoldOut();
    if (_publicPerWalletMaximum > 0 && getPublicSaleMinted(to) + quantity > _publicPerWalletMaximum) {
      revert InvalidQuantity();
    }
    if (_publicPerTransactionMaximum > 0 && quantity > _publicPerTransactionMaximum) revert InvalidQuantity();
    if (getPublicSaleMintedTotal() + quantity > getAvailablePublicSaleQuantity()) revert InvalidQuantity();
  }

  /**
   * @notice mint the specified number
   * @param quantity Number of mint
   * @dev Available for public sale only. Use presale() instead of this function during presale.
   */
  function mint(uint256 quantity) external payable {
    if (_saleStatus != SaleStatus.OPEN) revert SaleInactive();
    if (msg.value != _publicSalePrice * quantity) revert InvalidPrice();

    _validatePublicSaleMintQuantity(msg.sender, quantity);

    _safeMint(msg.sender, quantity);

    _increaseMintedCount(SaleStatus.OPEN, quantity);

    emit Minted(_currentIndex);
  }

  /**
   * @notice Mint NFT to the specified address
   * @param quantity Number of mint
   * @param recipient Recipient address
   */
  function airdrop(uint256 quantity, address recipient, SaleStatus countIncreaseSaleStatus) external onlyOwner {
    if (_currentIndex + (quantity - 1) > _totalSupply) revert SoldOut();
    if (countIncreaseSaleStatus == SaleStatus.CLOSED) revert InvalidSaleStatus();

    if (countIncreaseSaleStatus == SaleStatus.PRESALE) {
      _validatePreSaleMintQuantity(msg.sender, quantity);
    } else {
      _validatePublicSaleMintQuantity(msg.sender, quantity);
    }

    _safeMint(recipient, quantity);

    _increaseMintedCount(countIncreaseSaleStatus, quantity);

    emit AirDropped(_currentIndex);
  }

  /**
   * @notice get balance
   * @return balance
   */
  function getBalance() external view onlyOwner returns (uint256) {
    return address(this).balance;
  }

  /**
   * @notice Withdraw sales according to a specified distribution rate
   */
  function withdraw() external onlyOwner {
    uint256 balance = address(this).balance;

    for (uint256 i; i < _salesRecipients.length; i++) {
      _withdraw(_salesRecipients[i].recipient, (balance * _salesRecipients[i].percentages) / 100);
    }
  }

  /**
   * @notice Send money to the specified address and verify if it is successful
   * @param recipient Recipient address
   * @param amount amount of remittance
   */
  function _withdraw(address recipient, uint256 amount) internal {
    (bool success,) = recipient.call{value: amount}('');
    if (!success) revert WithdrawFailed();
  }

  function supportsInterface(bytes4 interfaceId) public view override(ERC721A, ERC2981) returns (bool) {
    return super.supportsInterface(interfaceId);
  }

  /**
   * @dev This contract is configured to use the DefaultOperatorFilterer, which automatically registers the
   *        token and subscribes it to OpenSea's curated filters.
   *        Adding the onlyAllowedOperator modifier to the transferFrom and both safeTransferFrom methods ensures that
   *        the msg.sender (operator) is allowed by the OperatorFilterRegistry. Adding the onlyAllowedOperatorApproval
   *        modifier to the approval methods ensures that owners do not approve operators that are not allowed.
   */
  function setApprovalForAll(address operator, bool approved) public override onlyAllowedOperatorApproval(operator) {
    super.setApprovalForAll(operator, approved);
  }

  function approve(address operator, uint256 tokenId) public override onlyAllowedOperatorApproval(operator) {
    super.approve(operator, tokenId);
  }

  function transferFrom(address from, address to, uint256 tokenId) public override onlyAllowedOperator(from) {
    super.transferFrom(from, to, tokenId);
  }

  function safeTransferFrom(address from, address to, uint256 tokenId) public override onlyAllowedOperator(from) {
    super.safeTransferFrom(from, to, tokenId);
  }

  function safeTransferFrom(
    address from,
    address to,
    uint256 tokenId,
    bytes memory data
  )
  public
  override
  onlyAllowedOperator(from)
  {
    super.safeTransferFrom(from, to, tokenId, data);
  }

  /**
   * @notice get presale max count
   * @return presaleSupply
   * @dev If the presale total supply is changed, the contract must be redeployed
   */
  function getPreSaleSupply() external pure returns (uint256) {
    return _presaleSupply;
  }

  /**
   * @notice Get Pre Sale Price
   * @return Public Sale Price
   */
  function getPreSalePrice() external view returns (uint256) {
    return _preSalePrice;
  }

  /**
   * @notice Set Pre Sale Price
   */
  function setPreSalePrice(uint256 price) external onlyOwner {
    _preSalePrice = price;
    emit UpdatedPreSalePrice(_preSalePrice);
  }

  /**
   * @notice Get the maximum number of mint allowed per wallet during a Pre sale
   * @return the maximum number of mint allowed per wallet
   */
  function getPrePerWalletMaximum() external view returns (uint256) {
    return _prePerWalletMaximum;
  }

  /**
   * @notice Set the maximum number of mint allowed per wallet during a Pre sale
   */
  function setPrePerWalletMaximum(uint256 perCount) external onlyOwner {
    _prePerWalletMaximum = perCount;
    emit UpdatedPrePerWalletMaximum(perCount);
  }

  /**
   * @notice Get the maximum number of mint allowed per transaction during a Pre sale
   * @return the maximum number of mint allowed per transaction
   */
  function getPrePerTransactionMaximum() external view returns (uint256) {
    return _prePerTransactionMaximum;
  }

  /**
   * @notice Set the maximum number of mint allowed per transaction during a Pre sale
   */
  function setPrePerTransactionMaximum(uint256 perCount) external onlyOwner {
    _prePerTransactionMaximum = perCount;
    emit UpdatedPrePerTransactionMaximum(perCount);
  }

  /**
   * @notice get pre sale sold supply
   */
  function getPreSaleMinted(address owner) public view returns (uint256) {
    return _preSaleMinted[owner];
  }

  /**
   * @notice Validate that the number of mints available in the presale has not been exceeded
   */
  function _validatePreSaleMintQuantity(address to, uint256 quantity) internal view {
    if (_currentIndex > _presaleSupply) revert SoldOut();
    if (_prePerWalletMaximum > 0 && getPreSaleMinted(to) + quantity > _prePerWalletMaximum) revert InvalidQuantity();
    if (_prePerTransactionMaximum > 0 && quantity > _prePerTransactionMaximum) revert InvalidQuantity();
    if (getPreSaleMintedTotal() + quantity > _presaleSupply) revert InvalidQuantity();
  }

  /**
   * @notice Mint the specified number on presale
   * @param quantity Number of mint
   */
  function presale(uint256 quantity) external payable {
    if (_saleStatus != SaleStatus.PRESALE) revert SaleInactive();
    if (msg.value != _preSalePrice * quantity) revert InvalidPrice();

    _validatePreSaleMintQuantity(msg.sender, quantity);

    _safeMint(msg.sender, quantity);

    _increaseMintedCount(SaleStatus.PRESALE, quantity);

    emit Minted(_currentIndex);
  }
}